

**A REVIEW OF THE 2006 SCOTTISH CRIME AND  
VICTIMISATION SURVEY: VICTIM FILE AND  
VICTIMISATION RATES**

**Paul Norris and Susan McVie**  
**Scottish Centre for Crime and Justice Research**  
**University of Edinburgh**

**July 2009**

## Introduction

This paper (and accompanying SPSS files) aims to address several concerns around the calculation of victimisation rates from the 2006 Scottish Crime and Victimization Survey (SCVS). Analysis of the original data revealed some problems with labelling of the data in the Victim Form dataset and some missing variables. Therefore, a review of the data and a check on the victimisation rates for 2006 was requested by Scottish Government. This review found that there were errors in the victimisation rates and estimates published in the report for the 2006 survey (Brown and Bolling 2007). This paper outlines the work done during the course of this review and presents a new set of victimisation rates and estimates that are comparable with previous crime surveys carried out in Scotland.

The revised dataset incorporates four key changes from the original 2006 SCVS dataset (released to the UK Data Archive on 8<sup>th</sup> January 2008 under study number SN5784):

1. Coding of offences in the 2006 survey used several “joint” crime codes which were not categorised in a way which was comparable with previous surveys. The revised dataset re-codes these “joint” offence codes that are comparable with the previous Scottish Crime Surveys.
2. Victimization rates for the 2006 survey should have been based on crimes which occurred between 1<sup>st</sup> April 2005 and 31<sup>st</sup> March 2006 (known as the reference period). However, an error in the syntax during construction of the victimisation rates meant that the derived variables included in the original dataset referred to crimes occurring between 1<sup>st</sup> July 2005 and 30<sup>th</sup> June 2006. The revised dataset has corrected this error.
3. In a small number of cases, victims reported that they could not recall exactly when the incident had occurred. The convention prior to 2006 was to include only those incidents where it was clear how many incidents had occurred during the reference period. In the 2006 data, where data were missing about the timing of series incidents, an average was taken across a number of time periods both in and outwith the reference period; however, this introduced error as to the number of incidents that had actually occurred in the reference period. The revised dataset excludes incidents which are not known to have occurred in the reference period.
4. Some of the variables from the original dataset have been relabelled for clarity and to facilitate secondary analysis of the dataset e.g. variables relating to the dates of incidents occurred. There is also a weighting variable to adjust for series incidents (capped at 5).

In addition to the revised datasets, the SPSS syntax used to create the new data files is provided to allow users to better understand how victimisation rates are calculated.

Generally speaking, the changes to the original dataset impact on a very small number of cases. However, the impact of these changes on victimisation rates varies between offences (i.e. rates do not consistently increase or decrease). In addition, it should be noted that when rates are aggregated to estimate the experience of the Scottish population changes to even a few cases can have a substantial impact on the estimated number of crimes.

## Differences between the Original Dataset and the Revised Edition<sup>1</sup>

### *Correction of Reference Period*

The documentation for the original 2006 SCVS dataset stated that victimisation rates and estimates were based on incidents occurring between 1<sup>st</sup> April 2005 and 31<sup>st</sup> March 2006; however, as a result of an error, incidents which had occurred between 1<sup>st</sup> July 2005 and 30<sup>th</sup> June 2006 were counted. This error resulted in the publication of victimisation rates and estimates that were incorrect. Table 1 shows the victimisation rates per 10,000 households (for household crime) and people (for personal crime) based on the original data, and with the reference period correctly calculated applied as twelve months from 1st April 2005. While several of the point estimates have changed, it is important to note that all the revised estimates are within the confidence intervals of the original estimates.

**Table 1: Victimisation rates per 10,000 households/people using the original 2006 SCVS data (July 05-June 06) and revised 2006 SCVS data (Apr 05-Mar 06)**

Crime Type	Original SCVS 2006 (95% CI)	Revised SCVS 2006 (95% CI) <sup>2</sup>	Difference
<b>COMPARABLE WITH THE POLICE</b>			
VANDALISM	<b>1,197</b>	<b>1,192</b>	<b>-5</b>
Motor Vehicle Vandalism	733 (632-833)	741 (638-842)	+7
Property Vandalism	464 (364-565)	452 (353-550)	-12
ACQUISITIVE	<b>424</b>	<b>433</b>	<b>+9</b>
House Breaking	282 (218-345)	294 (229-359)	+12
Theft of Motor Vehicle	45 (26-65)	45 (26-65)	0
Theft of Bicycle	97 (55-139)	94 (53-136)	-3
VIOLENCE	<b>765</b>	<b>667</b>	<b>-98</b>
Petty Assault	655 (470-839)	585 (443-727)	-70
Serious Assault	39 (9-69)	36 (10-61)	-3
Robbery	71 (18-123)	47 (16-77)	-24
<b>OTHER SURVEY CRIMES</b>			
Attempted Theft of/from Motor Vehicle	63 (38-89)	59 (35-83)	-4
Theft from Motor Vehicle	338 (270-405)	304 (241-366)	-34
Other Household Theft	592 (500-684)	565 (472-658)	-27
Theft from the Person	62 (30-95)	62 (30-95)	0
Other Personal Theft	304 (223-384)	302 (226-377)	-2
Petty Assault	655 (470-839)	585 (443-727)	-70
Serious Assault	39 (9-69)	36 (10-61)	-3
Motor Vehicle Vandalism	733 (632-833)	740 (638-842)	+6
Property Vandalism	464 (364-565)	452 (353-550)	-12
<b>ALL HOUSEHOLD CRIMES</b>	<b>2,615</b>	<b>2,553</b>	<b>-62</b>
<b>ALL PERSONAL CRIMES</b>	<b>1,131</b>	<b>1,031</b>	<b>-100</b>

<sup>1</sup> The syntax used to classify 'crime type', and to identify those cases which occurred within Scotland and in the twelve months from 1<sup>st</sup> April 2005, is provided in Appendix 3 (Syntax 1).

<sup>2</sup> Confidence intervals do not take account of clustered nature of data collection.

### *Dealing with Incomplete Date Information*

When a respondent reports that they have been a victim of a crime, they are asked for details of when the incident(s) occurred. Where they are reporting a series of related incidents, respondents are asked how many incidents occurred in each month in order to more accurately assess the timing of their victimisation. When a respondent cannot recall how many incidents occurred during specific months, they are asked to narrow it down to the nearest quarter. Where they cannot narrow it down to the nearest quarter, they are asked whether or not it occurred within the reference period (1<sup>st</sup> April 2005 to 28<sup>th</sup> March 2006). Historically, any incidents which could not be narrowed down to the reference period have not been included in the estimates of crime for the Scottish Crime Survey.

In the 2006 crime survey, however, these cases were handled differently. Where a victim reported the total number of crimes they had experienced, but they could not confirm how many of these incidents had occurred within the reference period, it was assumed that victimisation was spread evenly across time from the first to the last quarter in which the incidents could have occurred. For example, if a victim reported being a victim of four similar vehicle thefts occurring over a period from 1<sup>st</sup> January 2005 to 30<sup>th</sup> June 2005, it was assumed that 2 of these occurred in the first quarter (Jan-Mar) and 2 in the second quarter (Apr-Jun). This approach could lead to an over-counting of crimes (if 3 incidents occurred in Jan-Mar) or an under-counting of crimes (if 3 incidents occurred in Apr-Jun). The revised 2006 dataset includes only those cases where the exact number of incidents occurring within the reference period was known. It should be noted that, in the vast majority of cases, respondents were able to accurately report how many incidents occurred in each month or quarter within the reference period.

### *Differences in the Coding of Offences*

There are two main variables in the SCVS 2006 Victim Form dataset which indicate the type of crime that was reported to the survey. First, the variable 'offence' is a specific crime code based on the description of the victimisation provided by the respondent. This variable is coded in such a way as to be consistent with how the police would have recorded the incident. No change has been made to this variable between the original and revised datasets.

Once a specific crime code has been ascribed to each incident, these incidents are aggregated together to provide more general indicators of the level of crime by crime type. This variable is called "offtype" and it is this code which is used to create victimisation rates and estimates. During this review, it became apparent that some cases had been coded differently in 2006 to previous surveys. The main difference was the use of several new "joint" offence codes: "Serious Assault and Fire Raising", "Serious Assault and Housebreaking", "Rape and Housebreaking", and "Serious Assault with Sexual Motive and Housebreaking". As their titles suggest, these codes were used for victims who reported more than one type of crime within the same incident. Prior to 2006, such incidents took the code of the most serious type of victimisation (the coding manual included a priority list). The revised 2006 dataset reverts to this convention, and these offences are now coded as shown in Table 2.

**Table 2: Change in crime codes for ‘joint’ crimes**

<b>Original 2006 SCVS code</b>	<b>Revised 2006 SCVS code</b>
Serious Assault and Fire Raising	Serious Assault
Serious Assault and Housebreaking	Serious Assault
Rape and Housebreaking	Sexual Assault
Serious Assault with Sexual Motive & Housebreaking	Sexual Assault

The original 2006 dataset included one further crime category which was not present in previous surveys. The category “Motor Vehicle and Property Vandalism” contained any incident of fire-raising (irrespective of what was burned). Prior to 2006, these offences were separated depending on whether they involved a motor vehicle or other property, and then coded appropriately as either Vehicle Vandalism or Property Vandalism. There are a total of 13 eligible cases of fire-raising in SCVS 2006. Of these, four appear to relate to vehicle crime and nine to other forms of property. However, in the revised 2006 dataset all 13 cases are coded as property vandalism. While this marks a departure from the approach used in previous Scottish Crime Surveys, it reflects the coding which the Scottish Government intend to employ in the forthcoming SCJS which is comparable with the BCS approach.

The syntax provided by BMRB suggests that within the broad offence category of “Housebreaking” they had included the offences of Housebreaking In A Dwelling (Nothing Taken), Housebreaking In Dwelling (Something Taken), Attempted Housebreaking In A Dwelling, Attempted Housebreaking To A Non-Connected Domestic Garage/Or Shed, Housebreaking From A Non-Connected Domestic Garage/Outhouse (Nothing Taken) and Housebreaking From Non-Connected Domestic Outhouse (Something Taken). In previous surveys, “Housebreaking” was taken to include only the first tree of these offences with those offences involving non-connected properties been treated as examples of “Other Household Theft”. The coding of offences in the revised dataset reflects the codings employed prior to 2006 (see Table A1 in Appendix 1).

There are three further differences between the revised 2006 SCVS dataset and datasets from previous surveys. First, indecent exposure is now counted as a sexual offence; it was previous excluded from classification because sexual offences were taken to be those which involved contact with the respondent. Second, the definition of threats has being expanded to include threats made to the respondent which refer to a third person; previously, only threats which directly targeted at the respondent were included. Third, in previous surveys a distinction was drawn between those crimes which were “outside the scope of the survey” (i.e. occurred outside Scotland or outwith the reference period) and “other crimes” (crimes that were not covered by the survey). This distinction has being dropped for 2006. It is important to note that neither of these categories was used for calculating victimisation rates and so this change will not impact on any estimates produced.

Full details of which crimes are aggregated to crime types for the calculation of victimisation rates and estimates are given in Table A1 in Appendix 1. In addition, these individual crime categories are then further aggregated to represent more general crime categories e.g. ‘all violent crime’ or ‘all violence’. These crime categories are described in Table B1 in Appendix 1.

### *Series Weight Variables*

The original documentation for SCVS 2006 states,

“A Victim Form can represent either one incident or a series of similar and related incidents. Forms that represent a single incident got a weight of 1. Forms for a series of incidents were weighted by the number of incidents they represented. As in previous sweeps of the survey the weight for the series victim forms was capped at 5 – partly to limit the impact on victimisation rates of outliers and partly to limit the impact of the weights on the survey standard errors.” (Brown and Bolling, 2007, p38)

Due to an oversight, the series weighting variables were excluded from the original 2006 SCVS dataset. During this review, the series weights were recreated (using the revised data) and these are included in the revised Victim Form dataset. The capped weight is called “serieswt5” and the uncapped weight is called “serieswt”.

### **Victimisation Rates**

Table 3 presents another set of victimisation rates per 10,000 households (for household crime) and people (for personal crime) for the original SCVS 2006 dataset and the revised data. This differs from Table 1 in that it takes account of the revised offence classifications, discussed above. The 95% confidence intervals for the revised estimates all overlap with those from the original, which indicates that this minor recoding has had little impact.

**Table 3: Victimization rates per 10,000 households/people using the original 2006 SCVS data (July 05-June 06) and revised 2006 SCVS data (Apr 05-Mar 06 & new offence codes)**

<b>Crime Type</b>	<b>Original SCVS 2006 (95% CI)</b>	<b>Revised SCVS 2006 (95% CI)<sup>3</sup></b>	<b>Difference</b>
<b>COMPARABLE WITH THE POLICE</b>			
<b>VANDALISM</b>	<b>1,197</b>	<b>1,174</b>	<b>-23</b>
Motor Vehicle Vandalism	733 (632-833)	731 (630-833)	-2
Property Vandalism	464 (364-565)	443 (349-538)	-21
<b>ACQUISITIVE</b>	<b>424</b>	<b>433</b>	<b>+9</b>
House Breaking	282 (218-345)	197 (165-229)	-85
Theft of Motor Vehicle	45 (26-65)	45 (26-65)	0
Theft of Bicycle	97 (55-139)	94 (53-136)	-3
<b>VIOLENCE</b>	<b>765</b>	<b>651</b>	<b>-114</b>
Petty Assault	655 (470-839)	569 (429-710)	-86
Serious Assault	39 (9-69)	35 (10-61)	-4
Robbery	71 (18-123)	47 (16-77)	-24
<b>OTHER SURVEY CRIMES</b>			
Attempted Theft of/from Motor Vehicle	63 (38-89)	59 (35-83)	-4
Theft from Motor Vehicle	338 (270-405)	304 (241-366)	-34
Other Household Theft	592 (500-684)	662 (591-732)	+70
Theft from the Person	62 (30-95)	62 (30-95)	0
Other Personal Theft	304 (223-384)	299 (224-359)	-5
Petty Assault	655 (470-839)	569 (429-710)	-86
Serious Assault	39 (9-69)	35 (10-61)	-4
Motor Vehicle Vandalism	733 (632-833)	731 (630-834)	-2
Property Vandalism	464 (364-565)	443 (349-538)	-21
<b>ALL HOUSEHOLD CRIMES</b>	<b>2,615</b>	<b>2,537</b>	<b>-62</b>
<b>ALL PERSONAL CRIMES</b>	<b>1,131</b>	<b>1,013</b>	<b>-118</b>

### **Calculating Victimization Rates**

To calculate estimates of victimisation in Scotland it is necessary to aggregate the data from the 'crime' level to the 'individual' level. The Victim Form dataset which includes the variable 'offtype' is at the 'crime' level (i.e. each case relates to one single crime or series of crimes). To calculate victimisation rates per 10,000 people or households, it is necessary to aggregate the Victim Form data to the individual level (i.e. each case relates to one survey respondent) and to take account of those individuals who are not victims of crime. During the process of aggregation, each crime type becomes a separate variable. These derived variables have been appended to the master data file (using Syntax 2 in Appendix 3).

<sup>3</sup> Confidence intervals do not take account of clustered nature of data collection.

Before, creating victimisation rates, three decisions need to be made:-

1) Which crimes are you interested in?

Appendix 2 gives a list of the types of victimisation covered by the derived variables in the main data file. These variables cover both specific types of crime, e.g. Motor Vehicle Vandalism or House Breaking, and more general categories of offences, for example Acquisitive Crimes or All Personal Crimes.

Exactly how rates are created depends on whether the data you are analysing refers just to the respondent (personal crimes) or to the respondent and their household (property and vehicle related crimes).

2) Are you interested in prevalence or incidence?

For each type of victimisation you wish to estimate two variables are provided: these respectively indicate the prevalence of victimisation (been a victim at least once during the reference period) and a count of victimisation (how many times a respondent experienced victimisation during the reference period (capped at a maximum of 5). In the master data file, variables which indicate the count of offending end with an “n” while those indicating prevalence end in a “p”. For instance, the variable “noff1n” indicates the number of motor vehicle vandalism incidents experienced by each individual, while “noff1p” indicates whether or not each individual has experienced any incidents of motor vehicle vandalism. Full details of the appropriate variables for estimating the prevalence or count of victimisation for different types of crime are given in Appendix 2.

3) How should the data be presented?

The final decision to be made is how the data should be presented. Three main options are available:

1. Victimisation prevalence: The percentage of people in the survey (and, by implication, in the population) who have been a victim of crime.
2. Victimisation rate: The rate of victimisation per 10,000 of the population or per 10,000 households (or per 10,000 vehicle or bicycle owners).
3. Victimisation estimate: A grossed up figure for the total number of crimes estimated to have occurred across the whole of Scotland.

How the data are presented is dependent on the weight variable you apply during analysis. Table 4 shows the different weight variables available within the SCVS 2006 dataset and indicates the type of estimate which will be produced.

**Table 4: Weight variables in the Revised 2006 SCVS dataset**

<b>Weight Variable</b>	<b>Type of Estimate Provided</b>
ghh_wt	National estimates of victimisation for household based crimes. Use with variables noff1-noff8, vandal, acquis and allhh.
gpop_wt	National estimates of victimisation for respondent based (personal) crime. Use with variables noff9-noff15, violence and allpers.

hrate_wt	Rate of victimisation per 10,000 households. Use with variables noff1-noff8, vandal, acquis and allhh.
prate_wt	Rate of victimisation per 10,000 people. Use with variables noff9-noff15, violence and allpers.
veh_wt	Rate of victimisation per 10,000 vehicle owners. Use with variables noff3-noff5.
bike_wt	Rate of victimisation per 10,000 bicycle owners. Use with noff6.
Indwt	Prevalence of victimisation within the population.

### Examples of Syntax for Calculating Estimates<sup>4</sup>

#### *National Level of Household Crime (example Vandalism)*

A national estimate of the level of vandalism, involves looking at the number of incidents of vandalism that are estimated to have occurred across the whole of Scotland (i.e. the count measure is variable 'vandaln'). Vandalism is a household based crime, therefore, the data need to be weighted using the weight variable ghh\_wt. This gives the following SPSS syntax,

**Weight by ghh\_wt.**

**Fre vandaln /for not /sta sum.**

The above syntax produces the following output (a frequency table with the detail of the actual table hidden). The level of vandalism is represented by the value next to the word "Sum". In this case, the number of vandalism incidents is estimated at 375103.

#### Statistics

vandaln

N	Valid	2286660
	Missing	0
Sum		268662.38

#### *Rate of Household Crime per 10,000 Households (example Vandalism)*

Calculating the rate of vandalism per 10,000 households is identical to calculating the national estimate (above), but the data should be weighted using the variable hrate\_wt rather than ghh\_wt. This gives,

**Weight by hrate\_wt.**

**Fre vandaln /for not /sta sum.**

<sup>4</sup> The syntax presented in this section does not produce confidence intervals. The confidence intervals presented in this report were produced using the statistical package Stata.

*Prevalence Rate of Household Crime per 10,000 Households (example Vandalism)*

Estimating the prevalence of a crime (rather than the incidence) can be easily achieved by replacing the variable under investigation. For instance with regards to vandalism, the variable changes from 'vandaln' to 'vandalp' as shown below.

**Weight by hrate\_wt.  
Fre vandalp /for not /sta sum.**

*National Levels of Personal Crime (example Petty Assault)*

The number of incidents of petty assault suffered by each respondent to the crime survey is recorded in the variable 'noff9n'. This is a personal offence (meaning each respondent reports only their own experience, and not that of other people in their household). Therefore the data should be weighted using the variable 'gpop\_wt' to provide national estimates. The syntax for this reads,

**Weight by gpop\_wt.  
Fre noff9n /for not /sta sum.**

*National Prevalence of Personal Crime (example Petty Assault)*

An estimate of the percentage of people within the population who have experienced petty assault (the prevalence rate) can be created by changing the variable 'noff9n' to 'noff9p'.

**Weight by gpop\_wt.  
Fre noff9p /for not /sta sum.**

*Prevalence of Personal Crime per 10,000 People (example Petty Assault)*

For personal crimes, estimates per 10,000 people can be arrived at by weighting the data using variable 'prate\_wt'. For instance, the victimisation rate per 10,000 people for petty assault can be arrived at using the following syntax:

**Weight by prate\_wt.**  
**Fre noff9p /for not /sta sum.**

*National Level of Vehicle (or bicycle) crime (example Motor Vehicle Vandalism)*

National Estimates for Vehicle Crime, and theft of bicycles, are created by weighting the data to represent households, i.e. using 'ghh\_wt'. Therefore an estimate of the total number of motor vehicle vandalism (noff1n) incidents can be arrived at using the following syntax:

**Weight by ghh\_wt.**  
**Fre noff1n /for not /sta sum.**

*Rates of Vehicle (or Bike) Crime (example Motor Vehicle Vandalism)*

Rates of victimisation for vehicle crime can be presented in two ways. Firstly, they can be presented as a rate per 10,000 households (irrespective of whether or not a household has a vehicle). This is done using the weight variable

**Weight by hrate\_wt.**  
**Fre noff1n /for not /sta sum.**

Alternatively, rates can be presented per 10,000 households that own a vehicle. This is achieved using the variable veh\_wt as the weight variable. For instance the rate of motor vehicle vandalism per 10,000 households with vehicle owners can be estimated as follows:

**Weight by veh\_wt.**  
**Fre noff1n /for not /sta sum.**

Finally, a separate weight variable (bike\_wt) is available for calculating the rate of bicycle theft per 10,000 households with at least one bicycle. This gives the following syntax.

**Weight by bike\_wt.**  
**Fre noff6n /for not /sta sum.**

## Appendix 1: Details of which ‘crimes’ are aggregated to form ‘crime types’

Table A1: Individual crime types

Aggregated crime type (‘offtype’)	Individual crime codes (‘offence’)
Motor Vehicle Vandalism	Vandalism To A Motor Vehicle
Property Vandalism	Fire Raising, Vandalism To The Home (£20 Or Under), Vandalism To The Home (Over £20), Other Vandalism
Theft from Motor Vehicle	Theft From Car/Van, Theft From Motorbike/ Motorscooter Or Moped
Attempted Theft of/from a Motor Vehicle	Attempted Theft Of/From Car/Van, Attempted Theft Of/From Motorbike, Motorscooter Or Moped
Theft of Motor Vehicle	Theft Of Car/Van, Theft Of Motorbike, Motorscooter Or Moped
Bike Theft	Theft Of Pedal Cycle
Housebreaking	Housebreaking In A Dwelling (Nothing Taken), Housebreaking In Dwelling (Something Taken), Attempted Housebreaking In A Dwelling
Other Household Theft	Theft In A Dwelling, Theft From A Meter, Theft From Outside Dwelling (Excluding Theft Of Milk Bottles), Attempted Housebreaking To A Non-Connected Domestic Garage/Or Shed, Housebreaking From A Non-Connected Domestic Garage/Outhouse (Nothing Taken), Housebreaking From Non-Connected Domestic Outhouse (Something Taken)
Petty Assault	Minor Assault, Attempted Assault
Serious Assault	Serious Assault, Serious Assault And Fire Raising, Serious Assault And Housebreaking
Sexual Assault	Rape, Serious Assault With Sexual Motive, Assault With Sexual Motive, Attempted Rape, Indecent Assault, Indecent Exposure, Rape And Housebreaking, Serious Assault With Sexual Motive And Housebreaking
Theft from the Person	Snatch Theft From Person, Other Theft From Person, Attempted Theft From Person
Robbery	Robbery, Attempted Robbery
Other Personal Theft	Other Theft, Other Attempted Theft
Threats	Threat To Kill/Assault Made Against But Not Necessarily To The Respondent, Sexual Threat Made Against But Not Necessarily To The Respondent, Other Threat Or Intimidation Made Against But Not Necessarily To The Respondent, Threats Against Others, Made To The Respondent

The following offence codes are excluded from victimisation classification:

Theft Of Milk Bottles From Outside Dwelling, No Crime, Insufficient Information To Code The Offence, Crime Occurred Outside Scotland, Assault Falling Outside Survey's Coverage, Sexual Offence Outside The Survey's Coverage, Possibly Theft But Could Have Been A Loss/Possibly Attempted Theft, Other Robbery Or Theft From The Person Outside The Survey's Coverage, Possible Attempted Housebreaking, Other Housebreaking, Attempted Housebreaking, Theft In A Dwelling, Possible Theft/ Possible Lost Property, Other Theft/Attempted Theft Falling Outside Survey's Coverage, Possibly Criminal/Possibly Accidental Damage/Nuisance With Neighbours, Attempted Vandalism (No Damage Actually Received), Other Vandalism Falling Outside The Survey's Scope, Attempted Theft Falling Outside The Survey's Coverage, Threats/Intimidation Falling Outside The Survey's Coverage

Table B1: Aggregated crime types

<b>Aggregated crime category</b>	<b>Crime types</b>
Acquisitive Crime	Theft of Motor Vehicle, Bike Theft, Housebreaking
Violent Crime	Petty Assault, Serious Assault, Robbery
Vandalism	Motor Vehicle Vandalism, Property Vandalism
All Crimes Comparable with Police Statistics	Motor Vehicle Vandalism, Property Vandalism, Theft of Motor Vehicle, Bike Theft, Housebreaking, Petty Assault, Serious Assault, Robbery
All Survey Crimes	All Crimes in Table A1 except Threats and Sexual Assault
All Household Crimes	Motor Vehicle Vandalism, Property Vandalism, Theft from Motor Vehicle, Attempted Theft of/from a Motor Vehicle, Theft of Motor Vehicle, Bike Theft, Housebreaking
All Personal Crimes	Petty Assault, Serious Assault, Theft from the Person, Robbery, Other Personal Theft

## Appendix 2: Derived Victimization Variables in the Main File Dataset

Type of Victimization	Count Indicator	Prevalence Indicator
Motor Vehicle Vandalism	noff1n	noff1p
Property Vandalism	noff2n	noff2p
Theft from Motor Vehicle	noff3n	noff3p
Attempted Theft of/from a Motor Vehicle	noff4n	noff4p
Theft of Motor Vehicle	noff5n	noff5p
Bike Theft	noff6n	noff6p
Housebreaking	noff7n	noff7p
Other Household Theft	noff8n	noff8p
Petty Assault	noff9n	noff9p
Serious Assault	noff10n	noff10p
Sexual Assault	noff11n	noff11p
Theft from the Person	noff12n	noff12p
Robbery	noff13n	noff13p
Other Personal Theft	noff14n	noff14p
Threats	noff15n	noff15p
Acquisitive Crime	acquisn	acquisp
Violent Crime	violencn	violencep
Vandalism	vandaln	vandalp
All Crimes Comparable with Police Statistics	allcomn	allcomp
All Survey Crimes	allcrimen	allcrimep
All Household Crimes	allhhn	allhhp
All Personal Crimes	allpersn	allpersp

### Appendix 3: Syntax Used to Create Revised Data Files

#### 1. The syntax which is run on the original victim form data to code offences and identify eligible cases.

\*\*\*\*\*

SYNTAX FOR CREATING NEW ELIGIBLE VICTIM FILE AND COMPUTING  
VICTIMISATION RATES FOR 2006 SCVS.

Syntax written by Susan McVie and Paul Norris (June 2009)

Final decisions about which offences to include and how offences should be aggregated to crime categories based on discussion with the Scottish Government (8th June 2009)

Details of which cases should be excluded in terms of date based on discussion with the Scottish Government (8th June 2009)

Cases included where there is either a valid count for the number of offences in each quarter between April 2005 and March 2006 or where all incidents are reported to have happened during that time and a total number of incidents were available.

The relevant variables from the victim form file need to be aggregated to victim (individual) level and then imported into the main data file (Main A & B Combined)

\*\*\*\*\*

\*open original victim form file from UKDA and turn off all weights and selections

```
GET FILE="original victim forms.sav".  
weight off.  
FILTER OFF.  
USE ALL.  
EXECUTE .
```

\*\*\*\*\*

Start by calculating a new eligible variable (eligible) which excludes cases that are not eligible for victimisation analysis.

\*\*\*\*\*

\*Create eligibility variable which excludes those who didn't specifically say that the incident happened in Scotland.

```
compute eligwhere=0.  
if qloc=1 eligwhere=1.  
if qsco=1 eligwhere=1.  
val lab eligwhere 1 'eligible for analysis' 0 'not eligible'.  
var label eligwhere 'All variables eligible for analysis in victimisation rates: occurred in Scotland'.  
  
fre eligwhere.
```

\*Create eligibility variable for single incidents that happened within 2005/6 reference period.

```

compute eligsing=0.
if ((mthin ge 2) and (mthin le 13)) eligsing=1.
if ((qtrin ge 2) and (qtrin le 5)) eligsing=1.
var lab eligsing 'Eligible single incidents: those that occurred Apr 05-Mar 06 only'.
val lab eligsing 1 'eligible for analysis' 0 'not eligible'.

```

```
fre eligsing.
```

\*Create eligibility variable for series incidents where all of the incidents happened within 2005/6 financial year. This may be done using either prevalence (i.e the event occurred within the reference period

\*or a count of victimisation i.ee. how many incidence occurred in each quarter of the reference period

\*Using the 'dates' variables (prevalence of victimisation in ref period).

```

compute eligser1=0.
if (dates_2 = 1) or (dates_3 = 1) or (dates_4 = 1) or (dates_5 = 1) eligser1=1.
var lab eligser1 'Series incidents: those that occurred Apr 05-Mar 06 only'.
val lab eligser1 1 'eligible for analysis' 0 'not eligible'.

```

```
fre eligser1.
```

\*Using the 'nquart' variables (incidence of victimisation in ref period).

```

compute eligser2=0.
if (nquart1 ge 1) or (nquart2 ge 1) or (nquart3 ge 1) or (nquart4 ge 1) eligser2=1.
var lab eligser2 'Series incidents: those that occurred Apr 05-Mar 06 only'.
val lab eligser2 1 'eligible for analysis' 0 'not eligible'.

```

```
fre eligser2.
```

\*Calculating the number that occurred in the series within reference period (based on nquart).

```

recode nquart1 nquart2 nquart3 nquart4 (1 thru hi=copy) (else=0) into serq1 serq2 serq3
serq4.
do if eligser2=1.
compute eligser2n=serq1 + serq2 + serq3 + serq4.
end if.
var lab eligser2n 'Series incidents: number that occurred Apr 05-Mar 06 only'.

```

```
fre eligser2n.
```

\*Identifying those cases when exact number within quarter was not known, but all incidents occurred within the reference period

```

temp.
sel if (eligser1=1) and (eligser2 ne 1).
list vars=nquart1 nquart2 nquart3 nquart4 nquart5 nquart6 nquart7 nquart8 .

```

\*Identifying those cases that occurred only within reference period, although number in each period not known.

```

recode nquart5 nquart6 nquart7 (1 thru hi=copy) (-1=0) (else=sysmis) into serq5 serq6 serq7.
do if (eligser1=1) and (eligser2 ne 1).
compute eligserdk=1.
if (serq5 = 0) or (serq6 = 0) or (serq7 = 0) eligserdk=0.
end if.
var lab eligserdk 'Series incidents: resp said dk to when it happened, but all within 05/06
reference period'.
val lab eligserdk 1 'Within ref period' 0 'Some outwith ref period'.

fre eligserdk.

```

\*Create overall eligibility variable.

```

compute eligible=0.
if (eligwhere=1) and ((eligsing=1) or ((eligser2=1) or (eligserdk=1))) eligible=1.
var lab eligible 'All single and series incidents eligible within Scotland in the reference period
05/6'.
val lab eligible 1 'eligible' 0 'not eligible'.
exe.

```

```

*****
***Check number of eligible caes***

```

```

fre eligible.

```

```

**TOTAL OF 1521 CASES OCCUR WITHIN SCOTLAND AND WITHIN THE REFERENCE
PERIOD**

```

```

*****

```

```

*****

```

\*Create overall series weight for those cases eligible only.

```

*****

```

```

do if eligible=1.
compute serieswt=0.
if (pinci=2) serieswt=1.
if (eligser2=1) serieswt=eligser2n.
if (eligserdk=1) serieswt=nseries.
end if.
var lab serieswt 'All single and series incidents eligible: series incident weight (uncapped)'.

```

```

recode serieswt (0 thru 5=copy) (5 thru hi=5) into serieswt5.
var lab serieswt5 'All single and series incidents eligible: series incident weight (capped at 5)'.

```

```

fre serieswt serieswt5.

```

```

*****

```

Now compute a new offence variable (offtype) with correct classification for victims (i.e. correct housebreaking code).

Incidents of arson (13 cases) are all coded as household vandalism, rather than splitting between household (9 cases) and motor vehicle (4 cases)  
This removes the property/mv vandalism category from the previous 2006 coding.

Theft from a meter added to Other household theft (no cases in 2006)

Indecent exposure added to sexual assault

Other crimes and offences outwith the scope of the survey coded as one category (given code 16)

Serious assault and fire raising now coded as serious assault (as in 2003)

Serious assault and housebreaking now coded as serious assault (as in 2003)

Rape and housebreaking now coded as sexual assault (as in 2003)

Serious assault with sexual motive and housebreaking now coded as sexual assault (as in 2003)

Threats made to respondent referring to a third person now coded in Threats rather than Other Crimes (change from 2003)

\*\*\*\*\*

```
compute offtype=98.
if (offence=82) offtype=1.
if (offence=80) offtype=2.
if (offence=83 or offence=84 or offence=86) offtype=2.
if (offence=61 or offence=63) offtype=3.
if (offence=71 or offence=72) offtype=4.
if (offence=60 or offence=62) offtype=5.
if (offence=64) offtype=6.
if (offence=51 or offence=52 or offence=53) offtype=7.
if (offence=50 or offence=55 or offence=57 or offence=58 or offence=65 or offence=56)
offtype=8.
if (offence=12 or offence=21) offtype=9.
if (offence=11 or offence=14 or offence=15) offtype=10.
if (offence=31 or offence=32 or offence=33 or offence=34 or offence=35 or offence =36 or
offence=37 or offence=38) offtype=11.
if (offence=43 or offence=44 or offence=45) offtype=12.
if (offence=41 or offence=42) offtype=13.
if (offence=67 or offence=73) offtype=14.
if (offence=91 or offence=92 or offence=93 or offence=94) offtype=15.
if (offence=19 or offence=39 or offence=48 or offence=49
or offence=54 or offence=59 or offence=68 or offence=69 or offence=87 or offence=88
or offence=89 or offence=79 or offence=99) offtype=16.
if (offence=0 or offence=1 or offence=2 or offence=3 or offence=66 or offence=96 or
offence=97 or offence=98 ) offtype=16.

val lab offtype 1 'mv vandalism' 2 'property vandalism' 3 'theft from mv' 4 'att theft of/from mv'
5 'theft of motor vehicle' 6 'bicycle theft' 7 'housebreaking' 8 'other household theft'
9 'petty assault' 10 'serious assault' 11 'sexual assault' 12 'theft from person' 13 'robbery'
14 'other personal theft' 15 'threats' 16 'other crimes and incidents non-classifiable out or
outside the scope of the survey'.

var lab offtype 'Type of offence recorded for each incident of victimisation (2005/6)'.

fre offtype.
```

**\*\*Check offtype amongst incidents that are within Scotland and within April 2005-06 reference period\*\***

temp.  
sel if (eligible=1).  
fre offtype.

\*exclude incidents which did not involve crimes or were outwith scope of survey, or where crime is recorded in "Other crimes".  
\*(not used for victimisation rates).

if (offtype =16) eligible =0.  
exe.

\*\*\*\*\*  
**\*\*CHECK HOW MANY ELIGIBLE CASES REMAIN AFTER OFFENCE CODING**

fre eligible.

**\*\*NUMBER OF CASES REMAINING AFTER OFFENCE CODING IS 1418.  
\*\*THEREFORE 103 CASES HAVE BEEN REMOVED DUE TO EITHER BEEN OUT OF  
SCOPE OR CLASSIFIED IN THE "  
\*\*OTHER CRIMES" CATEGORY**

\*\*\*\*\*

\*\*\*\*\*  
The two weights ghh\_wt and gpop\_wt are contained only in the main data for 2005/6 data.

The data in the victim file need to be changed from one record for each victim form to one record per respondent with five variables per victim form .

Start by saving only a few necessary variables for the victim forms that are eligible into a new outfile (i.e. eligible=1) .

\*\*\*\*\*

sel if (eligible=1) .  
SAVE OUTFILE="SCVS 2006 Victim File aggregation.sav"  
/ keep serial vserial offtype serieswt5 .

GET FILE="SCVS 2006 Victim File aggregation.sav".

sort cases by serial.

rename variables (serieswt5 = serwt) .

CASESTOVARS  
/ID = serial  
/GROUPBY = VARIABLE .

\*\*\*\*\*

This assigns each of the five offence variables to one of the 15 single classification variables.

If you're not familiar with vectors, it's just a shorthand way of getting the same set of commands run multiple times - five in this case.

\*\*\*\*\*

\*CLASSIFICATION USING THE NEW SCS CATEGORIES .

```
vector noff(15) .
recode noff1 to noff15 (missing = 0) .
vector offtype = offtype.1 to offtype.5 .
vector series = serwt.1 to serwt.5 .
loop #i = 1 to 5 .
    if offtype (#i) = 1 noff1 = noff1 + series(#i) .
    if offtype (#i) = 2 noff2 = noff2 + series(#i) .
    if offtype (#i) = 3 noff3 = noff3 + series(#i) .
    if offtype (#i) = 4 noff4 = noff4 + series(#i) .
    if offtype (#i) = 5 noff5 = noff5 + series(#i) .
    if offtype (#i) = 6 noff6 = noff6 + series(#i) .
    if offtype (#i) = 7 noff7 = noff7 + series(#i) .
    if offtype (#i) = 8 noff8 = noff8 + series(#i) .
    if offtype (#i) = 9 noff9 = noff9 + series(#i) .
    if offtype (#i) = 10 noff10 = noff10 + series(#i) .
    if offtype (#i) = 11 noff11 = noff11 + series(#i) .
    if offtype (#i) = 12 noff12 = noff12 + series(#i) .
    if offtype (#i) = 13 noff13 = noff13 + series(#i) .
    if offtype (#i) = 14 noff14 = noff14 + series(#i) .
    if offtype (#i) = 15 noff15 = noff15 + series(#i) .
end loop .
exe.

rename variables (noff1 noff2 noff3 noff4 noff5 noff6 noff7 noff8 noff9 noff10 noff11 noff12
noff13 noff14 noff15
= noff1n noff2n noff3n noff4n noff5n noff6n noff7n noff8n noff9n noff10n noff11n noff12n
noff13n noff14n noff15n) .
exe.

var lab noff1n 'MV vandalism: number of incidents' / noff2n 'Property vandalism: number of
incidents' / noff3n 'Theft from MV: number of incidents'
/ noff4n 'Att theft of/from mv: number of incidents' / noff5n 'Theft of motor vehicle: number
of incidents' / noff6n 'Bicycle theft: number of incidents'
/ noff7n 'Housebreaking: number of incidents' / noff8n 'Other household theft: number of
incidents'
/ noff9n 'Petty assault: number of incidents' / noff10n 'Serious assault: number of incidents'
/ noff11n 'Sexual assault: number of incidents'
/ noff12n 'Theft from person: number of incidents' / noff13n 'Robbery: number of incidents'
/ noff14n 'Other personal theft: number of incidents'
/ noff15n 'Threats: number of incidents' .
exe.
```

\*\*\*\*\*

This takes the 15 classifications and creates Yes/No variables for each .

\*\*\*\*\*

```
recode noff1n noff2n noff3n noff4n noff5n noff6n noff7n noff8n noff9n noff10n noff11n noff12n
noff13n noff14n noff15n (1 thru hi = 1) into noff1p noff2p noff3p noff4p noff5p noff6p noff7p
noff8p noff9p noff10p noff11p noff12p noff13p noff14p noff15p.
```

## VARIABLE LABELS

noff1p 'Any motor vehicle vandalism incidents' /  
noff2p 'Any property vandalism incidents' /  
noff3p 'Any theft from motor vehicle incidents' /  
noff4p 'Any attempted theft of/from motor vehicle incidents' /  
noff5p 'Any theft of motor vehicle incidents' /  
noff6p 'Any bike theft incidents' /  
noff7p 'Any housebreaking incidents' /  
noff8p 'Any other household theft incidents' /  
noff9p 'Any petty assault incidents' /  
noff10p 'Any serious assault incidents' /  
noff11p 'Any sexual assault incidents' /  
noff12p 'Any theft from the person incidents' /  
noff13p 'Any robbery incidents' /  
noff14p 'Any other personal theft incidents' /  
noff15p 'Any threat incidents'.

recode noff1p to noff15p (missing = 2) .  
value labels noff1p to noff15p 1 "Yes" 2 "No" .

fre noff1p to noff15p.

\*Calculate derived indicators of victimisation for broad categories of offending\*  
\*First calculate count variables for the frequency of each type of crime\*

compute vandaln= noff1n+noff2n.  
compute acquisn=noff5n + noff6n + noff7n.  
compute violencen=noff9n + noff10n + noff13n.  
compute allcomn=noff1n + noff2n + noff5n + noff6n + noff7n + noff9n + noff10n + noff13n.  
compute allcrimen=noff1n + noff2n + noff3n + noff4n + noff5n + noff6n + noff7n + noff8n +  
noff9n + noff10n + noff12n + noff13n + noff14n.  
compute allhhn=noff1n + noff2n + noff3n + noff4n + noff5n + noff6n + noff7n + noff8n.  
compute allpersn=noff9n + noff10n + noff12n + noff13n + noff14n.

## VARIABLE LABELS

vandaln 'Number of vandalism incidents'/  
acquisn 'Number of acquisitive crime incidents' /  
violencen 'Number of violent crime incidents' /  
allcomn 'Number of all comparable crime incidents' /  
allcrimen 'Number of all crime incidents' /  
allhhn 'Number of household crime incidents' /  
allpersn 'Number of personal crime incidents' .

fre vandaln acquisn violencen allcomn allcrimen allhhn allpersn.

\*Use count measures as a basis for producing associated prevalence measures\*

recode vandaln acquisn violencen allcomn allcrimen allhhn allpersn (0=0) (1 thru hi=1) into  
vandalp acquisp violencep allcomp allcrimep allhhp allpersp.

## VARIABLE LABELS

vandalp 'Any vandalism incidents'/  
acquisp 'Any acquisitive crime incidents' /  
violencep 'Any violent crime incidents' /  
allcomp 'Any comparable crime incidents' /  
allcrimep 'Any crime incidents (all crimes)' /  
allhhp 'Any household crime incidents' /  
allpersp 'Any personal crime incidents' .

```
recode acquisp violencep allcomp allcrimep allhhp allpersp (0 = 2) .  
value labels acquisp violencep allcomp allcrimep allhhp allpersp 1 'Yes' 2 'No'.
```

```
fre vandalp acquisp violencep allcomp allcrimep allhhp allpersp.  
exe.
```

```
**Sort cases by serial and save respondent based data**
```

```
SORT CASES BY  
serial (A) .
```

```
SAVE OUTFILE="SCVS 2006 Victim File aggregation.sav"
```

## 2. Syntax for merging victimisation experience back on to the main data file.

```
*****  
*MERGE VICTIMISATION DATA BACK ON TO MAIN DATA FILE *  
*****
```

\*\*\*Open main data file without victimisation rate variables and sort cases by serial\*\*\*

```
GET  
FILE='main file without victimisation rates.sav'.  
SORT CASES BY  
serial (A) .
```

```
MATCH FILES /FILE=*  
/FILE='SCVS 2006 Victim File aggregation.sav'  
/BY serial.  
EXECUTE.
```

\*\*Save data file with victimisation rates\*\*

```
SAVE OUTFILE="SCVS 2006 main file with victim rates.sav".
```

\*\*Set cases without any valid victimisation to zero\*\*

```
RECODE  
noff1n noff2n noff3n noff4n noff5n noff6n noff7n noff8n  
noff9n noff10n noff11n noff12n noff13n noff14n noff15n (SYSMIS=0).  
EXECUTE .
```

```
RECODE  
noff1p noff2p noff3p noff4p noff5p noff6p noff7p noff8p  
noff9p noff10p noff11p noff12p noff13p noff14p noff15p (SYSMIS=0).  
EXECUTE .
```

```
RECODE  
vandaln acquisn violencen allcomn allcrimen allhhn allpersn (SYSMIS=0).  
EXECUTE .
```

```
RECODE  
vandalp acquisp violencep allcomp allcrimep allhhp allpersp (SYSMIS=0).  
EXECUTE .
```

```
RECODE  
noff1p noff2p noff3p noff4p noff5p noff6p noff7p noff8p  
noff9p noff10p noff11p noff12p noff13p noff14p noff15p  
vandalp acquisp violencep allcomp allcrimep allhhp allpersp (2=0).  
Execute.
```

```
val lab noff1p noff2p noff3p noff4p noff5p noff6p noff7p noff8p noff9p noff10p noff11p  
noff12p noff13p noff14p noff15p vandalp acquisp violencep allcomp allcrimep allhhp allpersp  
1 'Victim' 0 'Non-Victim' .  
EXECUTE.
```

```
fre noff1n noff2n noff3n noff4n noff5n noff6n noff7n noff8n  
noff9n noff10n noff11n noff12n noff13n noff14n noff15n.
```

```
fre noff1p noff2p noff3p noff4p noff5p noff6p noff7p noff8p  
noff9p noff10p noff11p noff12p noff13p noff14p noff15p.
```

**\*\*Save final data file with victimisation rates including derived variables for victimisation\*\***

SAVE OUTFILE="SCVS 2006 main file with victim rates.sav".